Hewlett-Packard

HP-19C/HP-29C SOLUTIONS

NAVIGATION



INTRODUCTION

This HP-19C/HP-29C Solutions book was written to help you get the most from your calculator. The programs were chosen to provide useful calculations for many of the common problems encountered.

They will provide you with immediate capabilities in your everyday calculations and you will find them useful as guides to programming techniques for writing your own customized software. The comments on each program listing describe the approach used to reach the solution and help you follow the programmer's logic as you become an expert on your HP calculator.

You will find general information on how to key in and run programs under "A Word about Program Usage" in the Applications book you received with your calculator.

We hope that this Solutions book will be a valuable tool in your work and would appreciate your comments about it.

TABLE OF CONTENTS

* THIS PROGRAM ALSO APPEARS IN THE HP-19C/29C APPLICATIONS BOOK. IT HAS BEEN INCLUDED HERE, IN SLIGHTLY MODIFIED FORM, FOR THE SAKE OF COMPLETENESS.

DISTANCE TO OR BEYOND HORIZON



This program computes the distance to an object of known height whose base is obscured by the horizon and whose top subtends a sextant altitude hs with the horizon. The sextant altitude is corrected for index error and height of eye. Additional features are the calculation of the distance to the horizon for a given height of eye and the distance of visibility of an object of height H above sea level.

EQUATIONS:

$$D = \sqrt{\left(\frac{\tan h_a}{2.46 \times 10^{-4}}\right)^2 + \frac{H-HE}{0.74736}} - \frac{\tan h_a}{2.46 \times 10^{-4}}$$

$$D_{hor} = 1.144 \sqrt{HE}$$

$$D_{vis} = 1.144(\sqrt{HE} + \sqrt{H})$$

where

D = distance to object, nautical miles

D_{hor}= distance to horizon, nautical miles

D_{vis}= distance of visibility, naut. miles

H = height of object beyond horizon, feet

HE= height of eye, feet

ha= $hs + IC - 0.97 \sqrt{HE}$

hs= sextant altitude, D.MS

IC= index correction, M.m

EXAMPLE 1:

The height of eye of of an observer is 9 feet above sea level, how far away is his horizon?

EXAMPLE 2:

An observer "bobs" Farallon Light on the horizon and finds his height of eye to be 16 feet. The light is 358 feet above sea level. How far is the observer from the light? (Accuracy is affected by abnormal refraction)

EXAMPLE 3:

The top of a lighthouse, whose base is obscured by the horizon, is known to be 300 feet above sea level. It is found to have a sextant altitude of 25!6 above the horizon. The height of eye is 20 feet and the sextant requires an index correction of +1!3.

What is the distance to the lighthouse?

What is the distance to the horizon?

It has been determined that the luminous range of the light is "strong", now compute its visibility for the given height of eye.

SOLUTIONS:

(1)

```
 9.00 ST02
      GSB2
 3.43  ***   n.m.
```

(2)

```
 16.00 ST02
358.00 ST03
       GSB2
        R/S
 26.22  ***  n.m.
```

(3)

```
  1.30 ST01
 20.00 ST02
300.00 ST03
0.2536 GSB1
  6.28  ***  n.m.
       GSB2
  5.12  ***  n.m.
        R/S
 24.93  ***  n.m.
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store pertinent data:				
	Index correction (minutes)	IC	STO	1	
	Height of eye (feet)	HE	STO	2	
	Height of object (feet)	H	STO	3	
3a.	Enter sextant height (D.MS) and compute distance to object	hs	GSB	1	D.(naut. mi.)
	and/or				
3b.	Compute distance to horizon and distance		GSB	2	D_{hor}(naut.mi.)
	of visibility		R/S		D_{vis}(naut.mi.)

Program Listings

01	*LBL1	hs(D.MS)
02	→H	hs°
03	RCL1	
04	RCL2	
05	√X	
06	.	
07	9	
08	7	
09	x	
10	-	
11	6	
12	0	
13	÷	
14	+	ha°
15	TAN	
16	2	
17	.	
18	4	
19	6	
20	EEX	
21	CHS	
22	4	
23	÷	
24	STO5	
25	RCL3	
26	RCL2	
27	-	
28	.	
29	7	
30	4	
31	7	
32	3	
33	6	
34	÷	
35	√X	
36	→P	$\sqrt{x^2 + y^2}$
37	RCL5	
38	-	
39	R/S	** D
40	*LBL2	
41	RCL3	
42	√X	
43	1	
44	.	
45	1	
46	4	
47	4	
48	x	1.144
49	LSTX	
50	RCL2	
51	√X	
52	x	
53	R/S	*** D_{hor}
54	+	
55	R/S	** D_{vis}

** "Printx" may be inserted before "R/S".

*** "Printx" may be used to replace "R/S".

REGISTERS

0	1 IC	2 HE	3 H	4	5 Used
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

DISTANCE BY HORIZON ANGLE AND DISTANCE SHORT OF HORIZON



This program calculates the distance between an observer and an object when (1) the vertical angle between its waterline and the horizon has been observed from a known height of eye or (2) the object's height is known, together with its subtended angle.

This program also calculates the height of an object if its subtended angle and distance from the observer are known.

EQUATIONS:

$$D = \frac{HE}{\tan(hs + IC + .97\sqrt{HE})}$$

$$D = \frac{H}{\tan(hs + IC)}$$

where

D = distance to object, feet

HE= height of eye, feet

IC= index correction, M.m

H = height of object, feet

hs= sextant altitude, D.MS

NOTE:

hs < 10' may make D unreliable due to atmospheric conditions when vertical sextant altitude between object and horizon is taken.

EXAMPLE 1:

The sextant altitude between the waterline of a buoy and the horizon is found to be 21!4. The observer has a height of eye of 22 feet and the sextant requires a +1!7 index correction. How far is the observer from the buoy?

EXAMPLE 2:

The sextant altitude subtended by the base and the top of a 41 foot light tower is 56!2. The sextant requires a -1!9 index correction. How far is the observer from the light tower?

EXAMPLE 3:

A vessel is anchored 2015 feet from an observer. The sextant altitude between the vessel's waterline and truck of mast is 1°15!2. There is no index error. How high is the truck of the mast above the waterline?

SOLUTIONS:

(1)

```
   1.70 STO1
  22.00 STO2
 0.2124 GSB1
2735.25  ***  ft.
         R/S
   0.45  ***  n.m.
```

(2)

```
  -1.90 STO1
  41.00 ENT↑
 0.5612 GSB2
2595.50  ***  ft.
         R/S
   0.43  ***  n.m.
```

(3)

```
   0.00 STO1
2015.00 ENT↑
 1.1512 GSB3
  44.08  ***  ft.
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store pertinent data:				
	Index correction (minutes)	IC	STO	1	
	Height of eye (feet)	HE	STO	2	
3a.	Enter sextant height (D.MS) and convert to	hs	GSB	1	D(feet)
	distance				
3b.	(Optional) convert to nautical miles		R/S		D(naut.mi.)
	OR				
4a.	Enter height of object (feet)	H	ENT↑		
	Enter sextant height (D.MS) and convert	hs	GSB	2	D(feet)
	to distance				
4b.	(Optional) convert to nautical miles		R/S		D(naut.mi.)
	OR				
5.	Enter distance to object (feet)	D	ENT↑		
	Enter sextant height (D.MS) and convert	hs	GSB	3	H(feet)
	to height				

Program Listings

```
01 *LBL1      hs
02   →H
03  RCL1
04  RCL2
05   √X
06    .
07    9
08    7
09    x
10    +
11  GSB0
12  RCL2
13   X⇄Y
14    ÷       D
15  GTO9
16 *LBL2      hs H
17   →H
18  RCL1
19  GSB0
20    ÷       D
21 *LBL9
22   R/S       ** D(ft.)
23    6
24    0
25    7
26    6
27    ÷
28   R/S       ** D(n.m.)
29 *LBL3      hs D(ft.)
30   →H
31  RCL1
32  GSB0
33    x
34   R/S       ** H (ft.)
35 *LBL0
36    6
37    0
38    ÷
39    +
40   TAN
41   RTN
```

** "Printx" may be inserted before "R/S".

REGISTERS

0	1 IC	2 HE	3	4	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

DISTANCE OFF AN OBJECT BY TWO BEARINGS

To determine the distance off an object as a vessel passes it, observe two bearings on the bow and note the distance run between bearings. The program calculates the distance off the object when it is abeam and at the time of the first and second bearings.



EQUATIONS:

$$D_2 = \frac{\sin RB_1}{\sin(RB_2 - RB_1)} D_{run}$$

$$D_{abeam} = |D_2 \sin RB_2|$$

$$D_1 = \left|\frac{D_{abeam}}{\sin RB_1}\right|$$

where

RB_1 = First relative bearing

RB_2 = Second relative bearing

D_{run} = St = Distance run

S = speed of vessel

t = time in minutes

D_1, D_2 = Distance at time of first or second bearing

D_a = Distance when abeam

EXAMPLE 1:

A lighthouse bears -026° (26° counter-clockwise) at 1130 and -051° at 1140. Our speed is 15 knots. How far will we be off the light when it is abeam? How far off were we at 1130 and 1140?

EXAMPLE 2:

A buoy is sighted bearing 015° on the bow, after a 3 mile run it bears 105°. What was its distance when abeam?

SOLUTIONS:

(1)

```
-26.00 ENT↑
-51.00 GSB1
 15.00 ENT↑
  0.10 GSB3
  4.60  ***   D1
        R↓
  2.59  ***   D2
        R↓
  2.02  ***   Dabeam
```

(2)

```
 15.00 ENT↑
105.00 GSB1
  3.00 GSB2
        R↓
        R↓
  0.75  ***   Dabeam
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store bearings (D.d):				
	at t_1	RB_1	ENT↑		
	at t_2	RB_2	GSB	1	
3a.	Enter distance run (naut. mi.)	R_{run}	GSB	2	D_i (naut.mi.)
	and compute distances				
	OR				
3b.	Enter speed (knots) and time (H.MS) and	S	ENT↑		
	compute distances	$t=t_2-t_1$	GSB	3	D_1 (naut.mi.)
4.	Display remaining distances:		R↓		D_2 (naut.mi.)
			R↓		D_a (naut.mi.)

Program Listings

01	*LBL1	Store bearings
02	STO2	
03	X⇄Y	
04	STO1	
05	R/S	
06	*LBL3	t,s
07	→H	
08	x	
09	*LBL2	D_{run}
10	RCL2	
11	RCL1	
12	-	
13	SIN	
14	÷	
15	RCL1	
16	SIN	
17	x	D_2
18	STO4	
19	RCL2	
20	SIN	
21	x	
22	ABS	D_{abeam}
23	RCL4	
24	LSTX	
25	RCL1	
26	SIN	
27	÷	D_1
28	ABS	D_1, D_2, D_{abeam}
29	R/S	

REGISTERS

0	1 RB_1	2 RB_2	3	4 Used	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

VELOCITY TRIANGLE

This program is an interchangeable solution for the vector addition problem. Given any two of the vectors shown, the program computes the third.



Compass course is corrected on input for magnetic variation and deviation. True course is decorrected on output to yield compass course. Remember to update the values used for variation (changes with location) and deviation (changes with heading).

EXAMPLE:

A vessel is making 2 knots through the water, steering 060° by the compass. The magnetic variation is 20.5°E and the deviation is 5°E. Calculate the set and drift of the current if the vessel is making good 3 knots on a course of 090°T.

SOLUTION:

```
 5.00 STO1
20.50 STO2
60.00 ENT↑
 2.00 GSB1
90.00 ENT↑
 3.00 GSB3
      GSB5
 1.02  ***    knots
      X⇄Y
98.86  ***    °T
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store compass corrections:				
	Deviation (negative if west)	±Dev°	STO	1	
	Variation (negative if west)	±Var°	STO	2	
3.	Enter any two of the three vectors:				
3a.	Heading -				
	Compass course	C_c°	ENT↑		
	Speed	S,knots	GSB	1	
3b.	Current -				
	Set	Set°	ENT↑		
	Drift	Drift,knots	GSB	2	
3c.	Course -				
	Course made good	CMG°	ENT↑		
	Speed made good	SMG,knots	GSB	3	
4.	Compute the remaining vector:				
4a.	Heading —				
	Speed		GSB	4	S,knots
	Compass course		x↔y		C_c°
	True course		RCL	4	C_t°
4b.	Current -				
	Drift		GSB	5	Drift/knots
	Set		x↔y		Set°
4c.	Course -				
	Speed made good		GSB	6	SMG,knots
	Course made good		x↔y		CMG°

Program Listings

*LBL1	Enter heading
STO5	
X⇄Y	
STO3	
RCL1	
+	
RCL2	
+	
GSB9	Normalize angle
STO4	
R/S	
*LBL2	Enter current
STO9	
X⇄Y	
STO8	
R/S	
*LBL3	Enter course
STO7	
X⇄Y	
STO6	
R/S	
*LBL4	Compute heading
GSB7	
RCL8	
RCL9	
CHS	
GSB0	
STO4	C_t
RCL1	
-	
RCL2	
-	
GSB9	Normalize angle
STO3	C_c
X⇄Y	
STO5	Speed
R/S	
*LBL5	Compute current
GSB7	
RCL4	
RCL5	
CHS	
GSB0	
STO8	Set
X⇄Y	
STO9	Drift
R/S	
*LBL6	Compute course
RCL4	
RCL5	
→R	x_1, y_1
ST.0	
X⇄Y	
RCL8	
RCL9	
GSB0	CMG
STO6	
X⇄Y	
STO7	SMG
R/S	
*LBL7	
RCL6	
RCL7	
→R	
ST.0	x_1, y_1
X⇄Y	
RTN	
*LBL0	x_2, y_2, y_1
→R	
S+.0	
R↓	
+	
RC.0	x_1+x_2, y_1+y_2
→P	
X⇄Y	
GSB9	Normalize angle
RTN	
*LBL9	
3	
6	
0	
→R	
→P	
X⇄Y	∠,360
X<0?	
GTO8	
X⇄Y	
R↓	∠
RTN	
*LBL8	
+	360 + ∠
RTN	

REGISTERS

0	1 Dev	2 Var	3 C_c	4 C_t	5 Speed
6 CMG	7 SMG	8 Set	9 Drift	.0 Used	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

COURSE TO STEER

This program calculates a course to steer given your location, the location where you want to go, your boat's speed through the water, and the set and drift of the current.



EXAMPLE:

A vessel making 6 knots through the water is at (45°N. 124°40'W) and she wishes to steer a course toward (44°40'N, 124°10'W). The magnetic variation is 20°.5E and there is a 2 knot current setting 090°. What course should she steer.

SOLUTION:

```
  0.00 ST01
 20.50 ST02
 90.00 ST08
  2.00 ST09
 45.00 ENT↑
124.40 ENT↑
 44.40 ENT↑
124.10 GSB1
 29.20  ***  Dist., n.m.
  6.00 GSB2
125.93  ***  Course to steer,degrees
        R/S
  7.30  ***  Speed made good,knots
        R/S
  4.00  ***  Transit time, H.MS
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store compass corrections:				
	Deviation (negative if west)	±Dev°	STO	1	
	Variation (negative if west)	±Var°	STO	2	
3.	Store current vector:				
	Set	Set°	STO	8	
	Drift	Drift/knots	STO	9	
4.	Enter positions and compute distance between:				
	Source latitude	L_1(D.MS)	ENT↑		
	Source longitude	λ_1(D.MS)	ENT↑		
	Destination latitude	L_2(D.MS)	ENT↑		
	Destination longitude	λ_2(D.MS)	GSB	1	Dist.(n.m.)
5.	Enter speed through the water and compute	S,knots	GSB	2	C_c°
	compass course to steer				
6.	Compute speed made good		R/S		SMG(knots)
7.	Compute time to reach destination		R/S		t(H.MS)

Program Listings

Step	Key	Comment	Step	Key	Comment
01	*LBL1	$\lambda_2 L_2 \lambda_1 L_1$	50	STO7	
02	→H		51	GSB7	SMG, CMB
03	X⇄Y	L_2	52	CHS	
04	→H		53	GSB0	
05	ST.1		54	RCL1	
06	R↓		55	-	
07	X⇄Y	λ_1	56	RCL2	
08	→H		57	-	
09	-		58	GSB9	C_c
10	CHS	$\lambda_1' - \lambda_2'$	59	R/S	
11	ST.2		60	RC.1	Dist
12	R↓		61	RCL7	
13	→H	L_1'	62	R/S	SMG (knots)
14	S-.1	$L_1' + L_2'$	63	÷	
15	+		64	→HMS	
16	2		65	R/S	Time (H.MS)
17	÷	Ang., latitude	66	*LBL9	Normalize:
18	COS		67	3	$0 < \text{angle} < 360$
19	RC.2		68	6	
20	x		69	0	
21	RC.1	$\sqrt{x^2 + y^2}$	70	→R	
22	→P		71	→P	
23	6		72	X⇄Y	
24	0	Convert to n.m.	73	X<0?	
25	x		74	GTO8	
26	ST.1		75	X⇄Y	
27	X⇄Y		76	R↓	
28	GSB9		77	RTN	
29	STO6	CMG	78	*LBL8	
30	X⇄Y		79	+	
31	R/S	Dist. (n.m.)	80	RTN	Vector add
32	*LBL2	Enter speed and	81	*LBL0	x_2, y_2
33	STO5	Compute C_c	82	→R	
34	RCL6		83	S+.0	
35	RCL8		84	R↓	
36	RCL6		85	+	$y_1 + y_2$
37	-		86	RC.0	$x_1 + x_2$
38	SIN		87	→P	
39	RCL9		88	X⇄Y	r
40	x		89	GSB9	θ
41	RCL5		90	RTN	θ'
42	÷		91	*LBL7	Vector add
43	SIN⁻¹		92	→R	x_1, y_1
44	-		93	ST.0	
45	GSB9	C_t	94	X⇄Y	
46	RCL5		95	RCL8	
47	GSB7		96	RCL9	
48	GSB0		97	RTN	
49	X⇄Y	SMG			

REGISTERS

0	1 Dev	2 Var	3	4	5 Speed
6 CMG	7 SMG	8 Set	9 Drift	.0 Used	.1 Used
.2 Used	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

ESTIMATED TIME OF ARRIVAL

This program computes the time of arrival at the next port in local zone time and GMT, when distance, speed, departure date and time in local zone time are input.

It also computes speed required to make a given ETA, when distance, date and time of departure in local zone time and date and time of desired arrival in local zone time are input.

All computations can be made in GMT by storing zeros in registers 4 and 5 and entering GMT times.

EXAMPLES:

1. A vessel departs San Francisco at 0030 on the 2nd of January local time, bound for Guam, distance 5,146 miles. What will be the date and time of arrival Guam local time, and GMT at 15.5 knots.

2. If the same Vessel makes the same departure time and wishes to arrive in Guam at 0700 on the 16th of January local Guam time, what speed is required.

NOTE:

Zone time of San Francisco is +8 and Guam is -10.

REFERENCE:

This program is adapted from HP-65 Users' Library program #02185A by Capt. Kenneth R. Orcutt.

SOLUTIONS:

(1)

```
5146.00 STO1
  15.50 STO2
   8.00 STO4
 -10.00 STO5
 2.0030 GSB1
16.1430  ***   Local time
            R/S
16.0430  ***   GMT
```

(2)

```
 2.0030 ENT↑
16.0700 GSB2
15.8582  ***   Knots
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store pertinent data:				
	Distance (nautical miles)	Dist	STO	1	
	Speed (knots)	Speed	STO	2	
	Departure time zone (negative if east)	DTZ	STO	4	
	Arrival time zone (negative if east)	ATZ	STO	5	
3.	To compute arrival time and date:				
3a.	Enter departure time and date in DD.HHMM	DD.HHMM	GSB	1	DD.HHMM (local time)
	format and run. e.g., 2:30 p.m. Jan 3				
	would be written 3.1430				
3b.	For GMT time and date of arrival		R/S		DD.HHMM(GMT)
	Note: If arrival is in the next month,				
	subtract the number of days in the				
	month.				
4.	To compute speed required to make a given				
	ETA:				
4a.	Enter departure time and date	DD.HHMM	ENT↑		
4b.	Enter arrival time and date and compute	DD.HHMM	GSB	2	Speed,knots
	speed				
	NOTE: If arrival is in the next month,				
	enter date plus the number of days				
	in the month.				

Program Listings

```
01 *LBL1      Compute arrival
02    4
03  STO0      i = 4
04    R↓
05  GSB0
06  RCL1
07  RCL2
08    ÷       Time of transit
09  RCL9
10    ÷
11    +
12  STO7      Add to departure
13  RCL5        time
14  RCL9
15    ÷
16    -       Arrival time
17 *LBL9
18   INT
19  LSTX      DD.
20   FRC
21  RCL9
22    x          Convert to hours
23  →HMS
24    1
25    0
26    0          Format display
27    ÷
28    +
29   R/S      **
30  RCL7      Arrival time and
31  GTO9        date, local and
32 *LBL0        GMT
33  FIX4      Set display
34   INT      DD.
35  LSTX
36   FRC
37    1
38    0
39    0
40    x
41   →H
42    2
43    4
44  STO9
45    ÷
46    +       Add DD.
47  RCLi      Time zone
48  RCL9
49    ÷
```

```
50    +
51   RTN
52 *LBL2      Arriv. time.Dep.
53    5         Time
54  STO0      i = 5
55    R↓
56  GSB0
57   X⇄Y      Dep. time
58   DSZ      i = 4
59  GSB0
60    -       Total transit time
61  RCL1
62   X⇄Y
63    ÷       Speed
64  RCL9
65    ÷       **
66   R/S      Speed, knots
```

** "PRINTX" may be inserted before "R/S".

REGISTERS

0 i	1 Distance	2 Speed	3	4 Dep.Time Zone	5 Arriv. TIME Zone
6	7 Used	8	9 24	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

GREAT CIRCLE NAVIGATION

This program calculates the great circle distance between two points and the initial course from the first point. Coordinates are input in degrees-minutes-seconds format. The distance is displayed in nautical miles and the initial course in decimal degrees.



EQUATIONS:

$$D = 60 \cos^{-1} \left[\sin L_1 \sin L_2 + \cos L_1 \cos L_2 \cos(\lambda_2 - \lambda_1)\right]$$

$$C = \cos^{-1}\left[\frac{\sin L_2 - \sin L_1 \cos(D/60)}{\sin(D/60)\cos L_1}\right]$$

$$C_i = \begin{cases} C; \sin(\lambda_2 - \lambda_1) < 0 \\ 360 - C; \sin(\lambda_2 - \lambda_1) \geqslant 0 \end{cases}$$

where:

L_1, λ_1 = coordinates of initial point

L_2, λ_2 = coordinates of final point

D = distance from initial to final point

C_i = initial course from initial to final point

REMARKS:

- Southern latitudes and eastern longitudes must be entered as negative numbers.
- Truncation and round off errors occur when the source and destination are very close together (1 mile or less).
- Do not use coordinates located at diametrically opposite sides of the earth.
- Do not use latitudes of +90° or -90°.
- Do not try to compute initial heading along a line of longitude ($L_1=L_2$).
- This program assumes the calculator is set in DEG mode.

EXAMPLE 1:

Find the distance and initial course for the great circle from Tokyo (L35°40'N, λ139°45'E) to San Francisco (L37°49'N, λ122°25'W).

EXAMPLE 2:

What is the distance and initial great circle course from L33°53'30"S, λ18°23'10"E to L40°27'10"N, λ73°49'40"W?

SOLUTIONS:

(1)

```
   35.40 ENT↑
 -139.45 ENT↑
   37.49 ENT↑
  122.25 GSB1
 4460.04  ***  (D,n.m)
          R/S
   54.37  ***  (Ci,dec.deg.)
```

(2)

```
-33.5330 ENT↑
-18.2310 ENT↑
 40.2710 ENT↑
 73.4940 GSB1
 6763.09  ***  (D,n.m.)
          R/S
  304.48  ***  (Ci,dec.deg.)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program.				
2.	Key in latitude and longitude of origin.	L_1(D.MS)	ENT↑		L_1
		λ_1(D.MS)	ENT↑		λ_1
3.	Key in latitude and longitude of	L_2(D.MS)	ENT↑		L_2
	destination.	λ_2(D.MS)			λ_2
4.	Calculate distance and initial course.		GSB	1	D(n.m.)
			R/S		C_i(dec.deg)

Program Listings

```
01 *LBL1
02   →H
03  STO3
04   R↓
05   →H
06  STO1
07   R↓
08   →H
09  STO4
10   R↓
11   →H
12  STO2
13  RCL2
14   SIN
15  RCL1
16   SIN
17    x
18  RCL2
19   COS
20  RCL1
21   COS
22    x
23  RCL3
24  RCL4
25    -
26  STO0
27   COS
28    x
29    +
30  STO5
31  COS⁻¹
32  STO6
33    6
34    0
35    x
36   R/S        ** D
37  RCL1
38   SIN
39  RCL2
40   SIN
41  RCL5
42    x
43    -
44  RCL2
45   COS
46    ÷
47  RCL6
48   SIN
49    ÷
50  COS⁻¹       C
51  RCL0
52   SIN
53  X<0?
54  GTO9
55   R↓
56    3
57    6
58    0
59   X⇄Y
60    -
61   RTN        ** Ci
62 *LBL9
63   R↓
64   RTN        ** Ci
```

** "Printx" may be inserted before "R/S" and "RTN".

REGISTERS

0 $\lambda_2 - \lambda_1$	1 L_2	2 L_1	3 λ_2	4 λ_1	5 COS D/60
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

GREAT CIRCLE COMPUTATION

This program computes the latitude corresponding to a specified longitude on a great circle passing through two given points.



EQUATIONS:

$$L_i=\tan^{-1}\left[\frac{\tan L_2 \sin(\lambda_i-\lambda_1)-\tan L_1 \sin(\lambda_i-\lambda_2)}{\sin(\lambda_2-\lambda_1)}\right]$$

where

(L_1,λ_1) = coordinates of initial point

(L_2,λ_2) = coordinates of final point

(L_i,λ_i) = coordinates of intermediate point

NOTES:

The program does not compute along lines of longitude ($\lambda_1 = \lambda_2$).

EXAMPLE:

A ship is proceeding from Manila to Los Angeles. The captain wishes to use great-circle sailing from L12°45!2N, λ124°20!1E, off the entrance to San Bernardino Strait, to L33°48!8N, λ120°07!1W, five miles south of Santa Rosa Island. Find the latitudes corresponding to 1) λ = 160°34'W; and 2) λ = 180°.

SOLUTION:

12.4512	ENT↑	
-124.2006	ENT↑	
33.4848	ENT↑	
120.0706	GSB1	
160.3400	GSB2	
41.2108	***	°N
180.0000	GSB2	
39.4133	***	°N

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter positions:				
	Initial latitude (negative for south)	L (D.MS)	ENT↑		
	Initial longitude (negative for east)	λ_1(D.MS)	ENT↑		
	Final latitude (negative for south)	L_2(D.MS)	ENT↑		
	Final longitude (negative for east)	λ_2(D.MS)	GSB	1	
3.	Enter intermediate longitude and compute	λ_i(D.MS)	GSB	2	L_i
	corresponding latitude				
4.	Repeat step 3 as desired.				

Program Listings

```
01 *LBL1
02   →H
03  STO3
04    R↓
05   →H
06  STO1
07    R↓
08   →H
09  STO4
10    R↓
11   →H
12  STO2
13   R/S
14 *LBL2
15   →H
16  STO8
17  RCL4
18     -
19   SIN
20  RCL1
21   TAN
22     x
23  RCL8
24  RCL3
25     -
26   SIN
27  RCL2
28   TAN
29     x
30     -
31  RCL3
32  RCL4
33     -
34   SIN
35     ÷
36  TAN⁻¹
37  →HMS
38   R/S
```

λ_2 L_2 λ_1 L_1

** λ_i

** L_i

** "Printx" may be inserted before "R/S".

REGISTERS

0	1 L_2	2 L_1	3 λ_2	4 λ_1	5
6	7	8 λ_i	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

COMPOSITE SAILING

When the great circle would carry a vessel to a higher latitude than desired, a modification of great-circle sailing, called composite sailing, may be used to good advantage. The composite track consists of a great circle from the point of departure and tangent to the limiting parallel, a course line along the parallel, and a great circle tangent to the limiting parallel and through the destination. This program computes, for each of two points, the longitude at which a great circle through the point is tangent to some limiting parallel.



EQUATIONS:

$$\lambda_{V1} = \lambda_1 + \cos^{-1}\left(\frac{\tan L_1}{\tan L_{max}}\right) s_1 s_2$$

$$\lambda_{V2} = \lambda_2 + \cos^{-1}\left(\frac{\tan L_2}{\tan L_{max}}\right) s_3 s_2$$

where

(L_1,λ_1) = initial position

(L_2,λ_2) = final position

(L_{max},λ_{V1}) = point at which limiting parallel is met

(L_{max},λ_{V2}) = point at which limiting parallel is left

$s_1 = \text{sgn}(\lambda_2 - \lambda_1)$

$s_2 = \text{sgn}(|\lambda_2 - \lambda_1| - 180)$

$s_3 = -s_1$

$$\text{sgn}(x) = \begin{cases} +1 \; ; & x \geqslant 0 \\ -1 \; ; & x < 0 \end{cases}$$

EXAMPLE:

A ship leaves Baltimore bound for Bordeaux (Royan), France. The captain desires to use composite sailing from L36°57!7N, λ75°42!2W one mile south of Chesapeake Light to L45°39!1N,λ1°29!8W, near the entrance to Grande Passe de l'Quest, limiting the maximum latitude to 47°N.

Required:

(1) The longitude at which the limiting parallel is reached.

(2) The longitude at which the limiting parallel should be left.

SOLUTION:

```
36.5742 ENT↑
75.4212 ENT↑
45.3906 ENT↑
 1.2948 GSB1
47.0000 GSB2
30.1607  ***   λV1 (D.MS)
          R/S
18.5653  ***   λV2 (D.MS)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter positions:				
	Initial latitude (negative for south)	L_1(D.MS)	ENT↑		
	Initial longitude (negative for east)	λ_1(D.MS)	ENT↑		
	Final latitude (negative for south)	L_2(D.MS)	ENT↑		
	Final longitude (negative for east)	λ_2(D.MS)	GSB	1	
3.	Enter latitude of limiting parallel	L_{max}(D.MS)	GSB	2	λ_{V1}(D.MS)
	and compute λ_{V1}(limiting parallel is met)				
4.	Compute λ_{V2}(limiting parallel is left)		R/S		λ_{V2}(D.MS)

Program Listings

```
01 *LBL1
02   →H
03 STO3        λ2  L2 λ1 L1
04   R↓
05   →H
06 STO1
07   R↓
08   →H
09 STO4
10   R↓
11   →H
12 STO2
13  R/S
14 *LBL2       Enter Lmax
15  TAN
16 STO5
17 RCL2
18  TAN
19  X⇄Y
20   ÷
21 COS⁻¹
22 RCL3
23 RCL4
24   -         λ2 - λ1
25 ENT↑
26  ABS        |λ2 - λ1| λ2 - λ1
27   1
28   8
29   0
30   -
31   x
32 ENT↑
33  ABS
34   ÷         ± 1
35 STO6
36  CHS
37   x
38 RCL4
39 *LBL0
40   +
41   1
42   →R        Normalize
43   →P        Angle
44  X⇄Y
45 →HMS
46  R/S        ** λv1,λv2
47 RCL1
48  TAN
49 RCL5
50   ÷
51 COS⁻¹
52 RCL6        sgn
53   x
54 RCL3
55 GTO0
```

** "PRINTX" may be inserted before "R/S".

REGISTERS

0	1 L_2	2 L_1	3 λ_2	4 λ_1	5 $\tan L_{max}$
6 sgn	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

RHUMB LINE NAVIGATION

This program is designed to assist in the activity of course planning. You supply the latitude and longitude of the point or origin and the destination. The program calculates the rhumb line course and the distance from origin to the destination.

Since the rhumb line is the constant-course path between points on the globe, it forms the basis of short distance navigation. In low and midlatitudes the rhumb line is sufficient for virtually all course and distance calculations which navigators encounter. However, as distance increases or at high latitudes the rhumb line ceases to be an efficient track since it is not the shortest distance between points.

The shortest distance between points on a sphere is the great circle. However, in order to steam great circles, an infinite number of course changes are necessary. Since it is impossible to calculate an infinite number of courses at an infinite number of points, several rhumb lines may be used to approximate a great circle. The more rhumb lines used the closer to the great circle distance the sum of the rhumb line distances will be. The Great Circle Computation program may be used to calculate intermediate course change points which can be linked by rhumb lines.

Latitudes and longitudes are input in degrees-minutes-seconds. Course is displayed in decimal degrees. Southern latitudes and eastern longitudes are input as negative numbers.



EQUATIONS:

$$C = \tan^{-1}\frac{\pi(\lambda_1 - \lambda_2)}{180(\ln\tan(45 + \tfrac{1}{2}L_2) - \ln\tan(45 + \tfrac{1}{2}L_1))}$$

$$D = \begin{cases} 60(\lambda_2 - \lambda_1)\cos L; \cos C = 0 \\ 60\,\dfrac{(L_2 - L_1)}{\cos C}; \text{ otherwise} \end{cases}$$

where:

(L_1, λ_1) = position of initial point

(L_2, λ_2) = position of final point

D = rhumb line distance

C = rhumb line course

REMARKS:

- No course should pass through either the south or north pole.
- Errors in distance calculations may be encountered as cos C approaches zero.
- Accuracy deteriorates for very short legs.
- This program assumes the calculator is set in DEG mode.

EXAMPLE 1:

What is the distance and course from L35°24'12"N,λ125°02'36"W to L41°09'12"N, λ147°22'36"E?

EXAMPLE 2:

What course should be sailed to travel a rhumb line from L2°13'42"S, λ179°07'54"E to L5°27'24"N, λ179°24'36"W? What is the distance?

SOLUTIONS:

(1)
```
     35.2412 ENT↑
    125.0236 ENT↑
     41.0912 ENT↑
   -147.2236 GSB1
     4135.60  ***    (DIST.,n.m.)
              R/S
      274.79  ***    (C,dec,deg.)
```

(2)
```
                +
     -2.1342 ENT↑
   -179.0754 ENT↑
      5.2724 ENT↑
    179.2436 GSB1
      469.31  ***    (DIST.,n.m.)
              R/S
       10.73  ***    (C,dec,deg.)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Key in latitude and longitude of origin	L_1(D.MS)	ENT↑		
		λ_1(D.MS)	ENT↑		
3.	Key in latitude and longitude of	L_2(D.MS)	ENT↑		
	destination	λ_2(D.MS)			
4.	Calculate distance and course		GSB	1	D(n.m.)
			R/S		C(dec.deg.)
	NOTE:				
	Southern latitudes and eastern longitudes				
	must be input as negative numbers.				

Program Listings

Step	Key	Comment	Step	Key	Comment
01	*LBL1		48	LN	
02	→H		49	RTN	
03	STO3	λ_2	50	*LBL8	
04	R↓		51	3	
05	→H		52	6	E to W 360 - C
06	STO1	L_2	53	0	
07	R↓	λ_1	54	RCL5	
08	→H		55	ABS	
09	STO4		56	-	
10	R↓		57	*LBL7	
11	→H		58	ABS	
12	STO2	L_1	59	STO6	
13	RCL4	$\lambda_1-\lambda_2$	60	1	
14	RCL3		61	8	
15	-		62	0	
16	STO0		63	RCL0	
17	2	Make $-180 < \lambda_1-\lambda_2$	64	ABS	
18	÷	< 180	65	X≤Y?	is $[\lambda_1-\lambda_2] > 180°$
19	SIN		66	GSB6	If so subtract from
20	SIN⁻¹		67	RCL1	360
21	9		68	COS	
22	0		69	×	
23	÷		70	STO7	
24	Pi		71	RCL1	
25	×		72	RCL2	
26	RCL1		73	-	
27	GSB9		74	RCL5	
28	RCL2		75	COS	
29	GSB9		76	X≠0?	is C = 90°?
30	-		77	÷	
31	→P		78	ENT↑	
32	R↓		79	X=0?	
33	STO5	C	80	RCL7	
34	RCL0		81	6	
35	SIN		82	0	
36	SIN⁻¹		83	×	
37	X<0?		84	ABS	
38	GTO8	x<0 means east to	85	R/S	** Distance
39	RCL5	west	86	RCL6	** Course
40	GTO7		87	RTN	
41	*LBL9	If west to east	88	*LBL6	
42	2	C is answer	89	3	
43	÷		90	6	If $[\lambda_1-\lambda_2] > 180°$
44	4		91	0	
45	5		92	X⇄Y	
46	+		93	-	then $360-[\lambda_1-\lambda_2]$
47	TAN		94	RTN	
			95	R/S	

REGISTERS

0 $\lambda_1-\lambda_2$	1 L_2	2 L_1	3 λ_2	4 λ_1	5 Used
6	7	8	9	.0	.1
.2	.3 ** "Printx" may be inserted before "R/S" & "RTN".				17
18	19	20	21	22	23
24	25	26	27	28	29

RHUMB LINE DEAD RECKONING



This program calculates a ship's DR position given the ship's course, speed, and elapsed time from the last fix or DR position. The DR position is stored so that on subsequent legs just course, speed, and elapsed time need be entered to obtain the updated DR position. The program may be used for both small and large area DR problems.

EQUATIONS:

The updated position (L,λ) is given by following a loxodrome (rhumb line) from the initial position (L_i,λ_i) for a distance determined by the speed and time.

$$L = L_i + \Delta t \frac{S \cos C}{60}$$

$$\lambda = \begin{cases} \lambda_i + \dfrac{180 \tan C(\ln\tan(45+\frac{L_i}{2}) - \ln\tan(45+\frac{L}{2}))}{\pi}; \\ \quad C \neq 90 \text{ or } 270 \ (L_i \neq L) \\ \lambda_i - \Delta t \dfrac{S \sin C}{60 \cos L_i}\ ;\ C=90 \text{ or } 270 \ (L_i = L) \end{cases}$$

where:

L_i = initial latitude (N, positive; S, negative)

L = updated latitude

λ_i = initial longitude (W, positive; S, negative)

λ = updated longitude

S = ship's speed, knots

C = ship's course, degrees

Δt = the time (H.MS) between initial and final positions.

NOTES:

1. The program cannot follow a meridian over a pole.
2. The program loses accuracy and gets incorrect answers when within 0.5° of a pole.

EXAMPLE (Fig. 1):

A vessel's position is L33°49!1N, λ120°52!0W at 1200. If she steams as shown, what is her position at each time?

TIME	C	S	DR
1200			L33°49'06"N,λ120°52'00"W
1330	120°	15 knots	(L33°37'51"N,λ120°28'34"W)
1510	240°	15 knots	(L33°25'21"N,λ120°54'32"W)
1823	90°	17 knots	(L33°25'21"N,λ119°49'01"W)
1955	355°	20 knots	(L33°55'54"N,λ119°52'14"W)

SOLUTION:

```
 33.4906 ENT↑
120.5200 GSB1
 13.3000 ENT↑
 12.0000 GSB3
120.0000 ENT↑
 15.0000 GSB2
 33.3751  ***
          R/S    1330 DR
120.2834  ***
 15.1000 ENT↑
 13.3000 GSB3
240.0000 ENT↑
 15.0000 GSB2
 33.2521  ***
          R/S    1510 DR
120.5433  ***
 18.2300 ENT↑
 15.1000 GSB3
 90.0000 ENT↑
 17.0000 GSB2
 33.2521  ***
          R/S    1823 DR
119.4902  ***
 19.5500 ENT↑
 18.2300 GSB3
355.0000 ENT↑
 20.0000 GSB2
 33.5554  ***
          R/S    1955 DR
119.5214  ***
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Enter initial position:				
	Latitude (negative for south)	L_i,D.MS	ENT↑		
	Longitude (negative for east)	λ_i,D.MS	GSB	1	
3.	Enter ship's way and compute DR:				
	Time on course*	Δt,H.MS	ENT↑		
	Course	C°	ENT↑		
	Speed	S,knots	GSB	2	L,D.MS
			R/S		λ,D.MS
*	To find the difference between two times	T_1,H.MS	ENT↑		
	T_1 may be less than or greater than T_2,	T_2,H.MS	GSB	3	Δt, H.MS
	but if the times are in different days,				
	add 24 hours to the smaller time.				

Program Listings

01	*LBL1	λ₁ L₁
02	FIX4	
03	→H	
04	STO2	
05	X⇄Y	
06	→H	
07	STO1	
08	R/S	
09	*LBL2	S C Δt
10	→R	
11	STO5	ScosC
12	R↓	
13	STO6	SsinC
14	R↓	
15	→H	
16	STO7	
17	RCL5	
18	x	
19	6	
20	0	
21	÷	
22	RCL1	L_i
23	+	
24	STO1	L
25	LSTX	L_i
26	X=Y?	
27	GTO9	C = 90° or 270°
28	GSB0	
29	RCL1	
30	GSB0	
31	÷	
32	LN	
33	RCL6	
34	x	
35	RCL5	
36	÷	
37	1	
38	8	
39	0	
40	x	
41	Pi	
42	*LBL8	
43	÷	
44	RCL2	λ_i
45	+	
46	1	
47	→R	Normalize Angle
48	→P	
49	X⇄Y	
50	STO2	λ
51	RCL1	
52	→HMS	
53	R/S	*** L
54	RCL2	
55	→HMS	
56	R/S	** λ
57	*LBL0	
58	9	
59	0	
60	+	
61	2	
62	÷	
63	TAN	
64	RTN	
65	*LBL9	
66	RCL6	
67	RCL7	
68	x	
69	RCL1	
70	COS	
71	6	
72	0	
73	x	
74	CHS	
75	GTO8	
76	*LBL3	Δt routine
77	→H	
78	X⇄Y	
79	→H	
80	-	
81	ABS	
82	→HMS	
83	R/S	** Δt

** "PRINTX" may be inserted before "R/S".

*** "PRINTX" may be used to replace "R/S".

REGISTERS

0	1 L	2 λ	3	4	5 ScosC
6 SsinC	7 Δt	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

CELESTIAL NAVIGATION AND DEAD RECKONING

This program allows you to update a vessel's position and correct it using sights on a celestial object. The program is started with your latitude and longitude and the object's GHA and declination all determined for the same time. Then when any other time is keyed in, the corresponding DR is computed. If a sight is taken at that time, the resulting altitude may be entered into the calculator to yield an intercept and azimuth. The DR may be moved accordingly if desired.

The dead reckoning technique used is mid-latitude sailing which, while not as accurate as rhumb line dead reckoning, is sufficiently good for most purposes. Altitude intercepts "toward" are considered to be positive, even though careful reading of Bowditch would indicate the opposite. By using this convention, it is easy to compute the intercept terminus (most probable position or MPP).

The program contains a useful subroutine, GSB 7, which can be used for translating almanac entries in degrees, minutes and tenths (DM.M) to decimal degrees (D.d).

REFERENCE:

This program is based on private communications with Paul E. Shaad of Sacramento, California.

EXAMPLE:

On February 19, 1975, a ship is steaming on course 240 at 17 knots. At 1800 GMT her dead reckoning position is 42°N,135°W. Compute her position at 2115.

Her navigator shoots the Sun from a height of 65' (dip = 7!8). At 2340 he obtains a sextant altitude of 28°25'36". Compute the altitude intercept and azimuth and correct the ship's DR.

SOLUTION

From The Nautical Almanac we take the Sun's GHA and declination at 1800 and 1900 GMT and also the Sun's semidiamter.

G.M.T.	SUN			
	G.H.A.		Dec.	
d h	°	'	°	'
19 18	86	31.5	S11	18.5
19	101	31.5		17.6
	S.D.	16.2		

```
  8631.5000 GSB7    GHA at 1800
    86.5250  ***
            ST01
 -1118.5000 GSB7    DEC at 1800
   -11.3083  ***
            ST02
    16.2000 GSB7    Semidiameter
     0.2700  ***
            ST03
    42.0000 ST04    Lat.            }
   135.0000 ST05    Long.           } Position
     7.8000 GSB7    Dip of the      }  at 1800
     0.1300  ***     horizon        }
            ST06
   240.0000 ST07    Course
    17.0000 ENT↑    Speed
    60.0000  ÷
     0.2833  ***    Speed converted to
            ST08     degrees per hour
    18.0000 ST09    Time
 10131.5000 GSB7
   101.5250  ***    } Calculation of
  8631.5000 GSB7    } Rate of Change
    86.5250  ***    } of GHA
              -
    15.0000  ***
            ST.1
 -1117.6000 GSB7
   -11.2933  ***    } Calculation of
 -1118.5000 GSB7    } Rate of Change
   -11.3083  ***    } of DEC
              -
     0.0150  ***
            ST.2
```

Now that the setup is compelte, you can dead reckon and reduce sights all day long.

```
  21.1500 GSB1  New time
  41.3223  ***  Latitude  }
           X⇄Y            }New Position at 2115
 136.0409  ***  Longitude }


  23.4000 GSB1  New time
  41.1150  ***  Latitude  }
           X⇄Y            }New Position at 2340
 136.5134  ***  Longitude }
  28.2536 GSB2  Sextant altitude
  -4.3689  ***  Altitude intercept
           X⇄Y
 219.4574  ***  Azimuth
           X⇄Y
          GSB7
  -0.0728  ***  Intercept converted to degrees
          GSB3
  41.1512  ***  Latitude  }
           X⇄Y            }Intercept terminus on Line of Position
 136.4752  ***  Longitude }
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Store the following values				
	Greenwich Hour Angle of object (negative if east)	GHA,D.d	STO	1	
	Declination of object (negative if south)	DEC,D.d	STO	2	
	Semi-diameter of object (negative if U.L.)	SD,D.d	STO	3	
	Latitude (negative if south)	L,D.d	STO	4	
	Longitude (negative if east)	λ,D.d	STO	5	
	Dip of the horizon	Dip,D.d	STO	6	
	Course	C,D.d	STO	7	
	Speed (in knots divided by 60)	S,D.d	STO	8	
	Time	t,H.h.	STO	9	
	Rate of change of GHA*	gha,D.d/hr	STO	1	
	Rate of change of dec*	dec.D.d/hr	STO	2	
3.	Enter a new time and compute new DR	t_{new}, H.MS	GSB	1	L,D.MS
			x↔y		λ,D.MS
4.	Enter sextant altitude and compute inter-	h_s, D.MS			
	cept and azimuth		GSB	2	a,mi.
			x↔y		Zn,D.d
5.	Update DR to MPP after GSB 2 in Step 4		GSB	7	a,D.d
	(i.e.: An in y; a in x)		GSB	3	L,D.MS
			x↔y		λ,D.MS
*	Enter GHA or DEC for some time	Value₁,D.MS	→H		Value₁,D.d
	Enter GHA or DEC for one hour earlier	Value₂,D.MS	→H		Value₂,D.d
	OR		-		Rate D.d/hr
	Enter GHA or DEC for some time	Value₁,DM.M	GSB	7	Value₁,D.d
	Enter GHA or DEC for one hour earlier	Value₂,DM.M	GSB	7	Value₂,D.d
			-		Rate D.d/hr

Program Listings


```
01 *LBL1
02    →H
03  RCL9
04   X⇄Y
05  STO9        Update
06   X⇄Y        Time
07    -
08  ST.4
09  RC.1
10    x         Update
11  ST+1        GHA
12  RC.4        And
13  RC.2
14    x         DEC
15  ST+2
16  RC.4
17  RCL8
18    x
19  RCL7
20   X⇄Y
21 *LBL3
22    →R        Update
23  ST+4
24    2         DR
25    ÷
26  RCL4
27   X⇄Y
28    -
29   COS
30    ÷
31  ST-5
32  RCL1
33  RCL5
34    -
35  STO0
36  RCL5
37  →HMS        New λ
38  RCL4
39  →HMS        New L
40   RTN
41 *LBL2        Compute
42    →H        a and Zn
43  RCL6
44    -
45  ENT↑
46   TAN
47   1/X
48    6
49    3
```

```
50    ÷
51    -
52  RCL3
53    +
54  ST.3
55  RCL0
56  RCL2
57   COS
58    →R
59  RCL4
60  ST.5
61   X⇄Y
62    →R
63  ST.4
64    R↓
65  RC.5
66  RCL2
67   SIN
68    →R
69  ST.5
70    R↓
71  RC.4
72    +
73  SIN-1
74  ST.4
75    R↓
76  RC.5
77    -
78    →P
79    R↓
80    1
81    8
82    0
83    +
84  RC.3
85  RC.4
86    -
87    6
88    0
89    x
90   RTN
91 *LBL7        Convert
92  →HMS        DM.m
93   EEX         to
94    2         D.d
95    ÷
96    →H
97   RTN
```

REGISTERS

0 LHA	1 GHA	2 DEC	3 SD	4 L	5 λ
6 Dip	7 C	8 S	9 t	.0	.1 gha
.2 dec	.3 H_o	.4 H_c	.5 Used	16	17
18	19	20	21	22	23
24	25	26	27	28	29

SIGHT REDUCTION TABLE

This program calculates the computed altitude, Hc, and azimuth, Zn, of a celestial body given the observer's latitude, L, and the local hour angle, LHA, and declination, d, of the body. It thus becomes a replacement for the nine volumes of HO 214. Moreover, the user need not bother with the distinctions of same name and contrary name; the program itself resolves all ambiguities of this type.

<u>EQUATIONS</u>:

$$Hc = \sin^{-1}[\sin d \sin L + \cos d \cos L \cos LHA]$$

$$Zn = \begin{cases} Z; & \sin LHA < 0 \\ 360-Z; & \sin LHA \quad 0 \end{cases}$$

$$Z = \cos^{-1}\left[\frac{\sin d - \sin L \sin Hc}{\cos L \cos Hc}\right]$$

<u>REMARKS</u>:

Southern latitudes and southern declinations must be entered as negative numbers.

The meridan angle t may be input in place of LHA, but if so, eastern meridan angles must be input as negative numbers.

The program assumes the calculator is set in DEG mode.

<u>NOTE</u>:

This program may also be used for star identification by entering observed azimuth in place of local hour angle and observed altitude in place of declination. The outputs are then declination and local hour angle instead of altitude and azimuth. The star may be identified by comparing this computed declination to the list of stars in <u>The Nautical Almanac</u>.

<u>EXAMPLE 1</u>:

Calculate the altitude and azimuth of the moon if its LHA is 2°39'54"W and its declination 13°51'06"S. The assumed latitude is 33°20'N.

<u>EXAMPLE 2</u>:

Calculate the altitude and azimuth of REGULUS if its LHA is 36°39'18"W and its declination is 12°12'42"N. The assumed latitude is 33°30'N.

<u>EXAMPLE 3</u>:

At 6:10 G.M.T. on January 12, 1977 a star peeked through the clouds over Corvallis (L44°34'N, 123°17'W). An alert observer using a bubble sextant quickly determined its altitude to be 26° and its azimuth 158°. Using the Nautical Almanac identify the star.

<u>SOLUTIONS</u>:

(1)

```
    33.20 ENT↑
 -13.5106 ENT↑
   2.3954 GSB1
  42.4447  ***   (Hc,D.MS)
           R/S
    183.5  ***   (Zn,dec.deg.)
```

(2)

```
  33.3000 ENT↑
  12.1242 ENT↑
  36.3918 GSB1
  50.2425  ***   (Hc,D.MS)
           R/S
    246.3  ***   (Zn,dec.deg.)
```

(3)

```
 44.3400 ENT↑
 26.0000 ENT↑
158.0000 GSB1
-16.3725  ***  (d,D.MS)
          R/S
   339.4  ***  (LHA,dec.deg.)
  123.17   →H
            +
   462.7  ***  (GHA,dec.deg.)
   203.4    -
         →HMS
   259.2  ***  (SHA,D.MS)
```

User Instructions

STEP	INSTRUCTIONS	INPUT DATA/UNITS	KEYS		OUTPUT DATA/UNITS
1.	Key in the program				
2.	Input the following:				
	Observer's latitude	L(D.MS)	ENT↑		
	Declination	d(D.MS)	ENT↑		
	Local hour angle	LHA (D.MS)			
3.	Calculate:				
	Altitude		GSB	1	Hc(D.MS)
	Azimuth		R/S		Zn(dec.deg.)
	<u>OR</u>:				
2.	Input:				
	Observer's latitude	L(D.MS)	ENT↑		
	Altitude	Hc(D.MS)	ENT↑		
	Azimuth	Zn(D.MS)			
3.	Calculate:				
	Declination		GSB	1	d(D.MS)
	Local hour angle		R/S		LHA (dec.deg.)

Program Listings


```
01 *LBL1
02    →H
03  STO2        L
04    R↓
05    →H
06  STO1        d
07    R↓
08    →H
09  STO0        LHA
10   SIN
11  RCL1
12   SIN
13    x
14  RCL0
15   COS
16  RCL1
17   COS
18    x
19  RCL2
20   COS
21    x
22    +
23  STO3
24  SIN⁻¹
25  STO4        Hc,dec.deg.
26  →HMS        Hc,D.MS
27  FIX4        ***
28   R/S
29  FIX1
30  RCL1
31   SIN
32  RCL3
33  RCL0
34   SIN
35    x
36    -
37  RCL0
38   COS
39    ÷
40  RCL4
41   COS
42    ÷
43  COS⁻¹
44  RCL2
45   SIN        Z
46  X<0?
47  GTO0
48    R↓
49    3
50    6
51    0
52   X⇄Y
53    -         ** Zn
54   RTN
55 *LBL0
56    R↓        ** Zn
57   RTN
58   R/S
```

** "PRINTX" may be inserted before "RTN".

*** "PRINTX" may be used to replace "R/S".

REGISTERS					
0 L	1 d	2 LHA	3 Sin Hc	4 Hc	5
6	7	8	9	.0	.1
.2	.3	.4	.5	16	17
18	19	20	21	22	23
24	25	26	27	28	29

In the Hewlett-Packard tradition of supporting HP programmable calculators with quality software, the following titles have been carefully selected to offer useful solutions to many of the most often encountered problems in your field of interest. These ready-made programs are provided with convenient instructions that will allow flexibility of use and efficient operation. We hope that these Solutions books will save your valuable time. They provide you with a tool that will multiply the power of your HP-19C or HP-29C many times over in the months or years ahead.

Mathematics Solutions
Statistics Solutions
Financial Solutions
Electrical Engineering Solutions
Surveying Solutions
Games
Navigational Solutions
Civil Engineering Solutions
Mechanical Engineering Solutions
Student Engineering Solutions